Today’s Lecture COS 333, Spring ’96

1. Projects

2. Finish up tcl/tk

3. Visual Basic, interactively

ffl Hello, World ffl A brief look at couple of MS demos (a) simple calculator (b) simple text editor ffl A detailed look at a VB application, Speed-Reader (SR) speed-reader.bas COS 333, Spring ’96

’ speed-reader.bas
’ displays, in rapid succession,
’ words from a text file

Sub Command1_Click ()
    toggle
    HScroll1_Change
    Timer1.Enabled = True
End Sub

Sub Command2_Click ()
    toggle
    Timer1.Enabled = False
End Sub

Sub finishRead ()
    Command2.Enabled = False ’pause
    Command1.Enabled = True ’run
End Sub

Sub toggle ()
    Command1.Enabled = Reading
    Command2.Enabled = Not Reading
    Reading = Not Reading
End Sub

Sub Form_Load ()
    Timer1.Enabled = False
End Sub

Sub initRead ()
    wordsToGo = allLines
    thisWord = 1
End Sub

speed-reader.bas (cont’d) COS 333, Spring ’96

Sub mainOpen_Click ()
    Static thisWord As Integer
    On Error GoTo ErrHandler
    CMDialog1.Filter = "All files (*.*)|*.*|Text files (*.txt)|*.txt| \
Doc files (*.doc)|*.doc"
    CMDialog1.FilterIndex = 2
    CMDialog1.Action = 1
    Open CMDialog1.Filename For Input As 1 Len = 4096
    On Error GoTo ErrHandler2
    allLines = ""
    Screen.MousePointer = 11 ’hourglass
    Do Until EOF(1)
        Line Input #1, NextLine
        allLines = allLines + NextLine + " "
    Loop
    Close 1
    Screen.MousePointer = 0 ’default
    Command1.Enabled = True
    Label3.Caption = CMDialog1.Filename
    Reading = False
    wordsRead = 0
    Label5.Caption = wordsRead
    initRead
    Exit Sub
ErrHandler:
    msg = "Error opening " + CMDialog1.Filename
    MsgBox msg, MB_ICONEXCLAMATION, "Error"
    Exit Sub
ErrHandler2:
    Close 1
    Screen.MousePointer = 0 ’default
    msg = "Error reading " + CMDialog1.Filename
    MsgBox msg, MB_ICONEXCLAMATION, "Error"
    Exit Sub
End Sub

speed-reader.bas (cont’d) COS 333, Spring ’96

Sub mainExit_Click ()
    End
End Sub

Sub Spin1_SpinDown ()
    If HScroll1.Value > HScroll1.Min Then
        HScroll1.Value = HScroll1.Value - 1
        HScroll1_Change
    End If
End Sub

Sub Spin1_SpinUp ()
    If HScroll1.Value < HScroll1.Max Then
        HScroll1.Value = HScroll1.Value + 1
        HScroll1_Change
    End If
End Sub

Sub Text1_Change ()
    If IsNumeric(Text1.Text) Then
        v = Val(Text1.Text)
        If v >= HScroll1.Min And v <= HScroll1.Max Then
            HScroll1.Value = Text1.Text
            HScroll1_Change
        End If
    End If
End Sub

Sub HScroll1_Change ()
    Text1.Text = HScroll1.Value
    Timer1.Interval = 1000# / (HScroll1.Value / 60#)
End Sub

speed-reader.bas (cont’d) COS 333, Spring ’96

Sub Timer1_Timer ()
    If thisWord = 0 Then
        Command2_Click
        initRead
        wordsRead = 0
    Else
        done = False
        While Not done
            If thisWord > 0 Then
                nextBrk = InStr(thisWord, allLines, " ")
                If nextBrk = 0 Then
                    thisWord = 0
                ElseIf nextBrk <> thisWord Then
                    word$ = Mid$(allLines, thisWord, nextBrk - thisWord)
                    word$ = LTrim$(RTrim$(word$))
                    thisWord = nextBrk + 1
                    If word$ <> "" Then
                        Picture1.Cls
                        Picture1.Print word$
                        wordsRead = wordsRead + 1
                        Label5.Caption = wordsRead
                        done = True
                    End If
                Else
                    thisWord = thisWord + 1
                End If
            Else
                done = True
            End If
        Wend
    End If
End Sub

’ end of speed-reader.bas

Next Lecture COS 333, Spring ’96

Java (check out the Java link on the homepage)